
Le  Journal  du  Soft

LA  ROM  DEPLOMBEUSE
Par André Chéramy et Claude Sittler

Cette ROM, qui date de 1986 devrait vous aider à régler un problème qui est resté d’actualité, comme
nous l’avons vu dans notre article précédent. Les auteurs sont partis de la première mouture de la
ROM V1.1. Mais nous vous conseillons de partir de la seconde, le résultat n’en sera que meilleur. Les
changements effectués sont peu nombreux, mais cruciaux. Il s’agit :
1) De l’ajout d’une commande OLD (les utilisateurs de K7 ne disposaient pas de Sedoric).
2) De la neutralisation totale du démarrage AUTO des programmes chargés.
3) De la suppression de la routine qui « vérifie » la RAM. C’est l’effet de peigne qui écrase toutes les
cases memoires, en écrivant des #AA (10101010) puis des #55 (01010101), les fameux «UUUU». Or
la protection la plus courante des programmes de cette époque consistait à provoquer un RESET de
type « Cold Start », lequel effectue entre autres choses, une « vérification » de la RAM…
4) De la modification, accessoire mais bien pratique, du message de Copyright (sinon, dans quelques
temps, vous oublierez la nature de la ROM présente dans votre Atmos).

DETAIL DES MODIFICATIONS APPORTEES

Cette ROM sera distribuée avec la disquette trimestrielle de décembre 2001, mais vous pouvez aussi
la fabriquer vous-même en suivant la procédure que nous avons déjà décrite à mainte reprise dans
notre série d’articles du « Journal du Soft ». En partant de la ROM V1.1 dernière mouture, dont la
checksum est #1A25, la checksum du fichier produit DEPLOMBE.ROM devra être #26E3. En voici
le détail en suivant l’ordre croissant des adresses :

En #C032 et #C033, adresse d’exécution de la commande LET. Rappelons que dans la table des
adresses d’exécution sont portés deux octets pour chaque commande : un octet de poids faible, puis
un octet de poids fort, formant ensemble l’adresse –1 de la commande.  Ici, les octets #1B et #CB pour
la commande LET située en #CB1C ont été modifiés en #13 et #FA pour accéder à la nouvelle routine
OLD implantée en #FA14 (voir le pourquoi de ceci plus loin).

De #C149 à #C14B, le nom de la commande LET (soit #4C, #45 et #D4, le code ASCII du dernier
caractère étant augmenté de #80, ici « T » = #54 + #80 = #D4) a été remplacé par celui de la com-
mande OLD (#4F, #4C, #C4). Remarquez que la longueur est identique. La commande LET n’est
jamais utilisée dans les programmes, c’est une commande implicite qui est lancée par l’interpréteur
lorsqu’il rencontre le signe « = ». Notez que la routine correspondante n’a pas été touchée, puisqu’elle
est toujours utilisée par l’interpréteur. On a seulement récupéré la place du nom. Un appel à la com-
mande OLD mettra donc en jeu la routine située en #FA14, tandis que le signe « = » continuera à
utiliser la routine en #CB1C.

Dans la routine principale de la commande CLOAD, sise de #E85B à #E908, les zones traitant du
lancement automatique de #E8DB à #E8E7, pour les programmes en langage machine et de #E8F6 à
#E902, pour les programmes BASIC, ont été supprimées. En fait ces zones ont été remplies de #EA
(pour NOP, No Operation). En conséquence, aucun programme ne s’exécutera plus automatiquement.

De #ED9B à #EDBB, le message de Copyright « ORIC EXTENDED BASIC V1.1 © 1983 TANGE-
RINE » a été modifié en  « ORIC DEPLOMBEUSE V1.1 © 1986 MARGARINE »

En #F915 et #F91A, les couleurs de INK (#00 noire) et PAPER  (#17 blanc) ont été changées en #07
(encre blanche) et #14 (papier bleu).

De #FA14 à #FA84, la routine de vérification de la RAM à été remplacée par la routine OLD suivante :

FA14-   A9 00     LDA #$00     Ecrit l’adresse #9800
FA16-   85 A6     STA $A6      dans le pointeur HIMEM
FA18-   A9 98     LDA #$98     en #A6-#A7
FA1A-   85 A7     STA $A7
FA1C-   78        SEI          Interdit les interruptions
FA1D-   A0 01     LDY #$01     Ecrit l’adresse #0500 en #00-#01
FA1F-   84 02     STY $02      Ce sera l’adresse où lire un octet
FA21-   A0 00     LDY #$00     Ecrit l’adresse #0501 en #02-#03
FA23-   84 00     STY $00      Ce sera l’adresse où copier le lien
FA25-   A9 05     LDA #$05     manquant, au début de la 1ere ligne
FA27-   85 01     STA $01
FA29-   85 03     STA $03      L’index Y vaut encore #00
FA2B-   B1 00     LDA ($00),Y  Lit l’octet en #0500
FA2D-   D0 4E     BNE $FA7D    Quitte si pas nul (#00 = flag BASIC)
FA2F-   A5 00     LDA $00
FA31-   18        CLC
FA32-   69 04     ADC #$04     Ajoute 4 au pointeur en #00-#01
FA34-   85 00     STA $00      qui passe donc à #504, adresse
FA36-   A5 01     LDA $01      du dernier octet du n° de ligne
FA38-   69 00     ADC #$00     donc juste avant le 1er octet du
FA3A-   85 01     STA $01      code proprement dit
FA3C-   20 52 FA  JSR $FA52    Incrémente le pointeur en #00-#01
FA3F-   B1 00     LDA ($00),Y  recherche le zéro qui marque la fin
FA41-   D0 F9     BNE $FA3C    de la première ligne
FA43-   20 52 FA  JSR $FA52    Trouvé : Incrémente le pointeur
FA46-   A5 00     LDA $00      Lit dans le pointeur l’adresse du
FA48-   91 02     STA ($02),Y  1er octet de la deuxième ligne
FA4A-   C8        INY          et l’écrit à sa place normale
FA4B-   A5 01     LDA $01      au début de la 1ere ligne, c’est
FA4D-   91 02     STA ($02),Y  le lien manquant, effacé par NEW
FA4F-   4C 59 FA  JMP $FA59    By-passe le sous-programme :
FA52-   E6 00     INC $00      Sous-programme d’incrémentation
FA54-   D0 02     BNE $FA58    du pointeur #00-#01
FA56-   E6 01     INC $01
FA58-   60        RTS          Fin du sous-programme
FA59-   20 52 FA  JSR $FA52    Incrémente le pointeur en #00-#01
FA5C-   A0 00     LDY #$00     Recherche l’adresse de fin du BASIC
FA5E-   B1 00     LDA ($00),Y  c’est à dire l’adresse de l’octet
FA60-   D0 1D     BNE $FA7F    qui suit les trois zéros finaux
FA62-   C8        INY
FA63-   B1 00     LDA ($00),Y
FA65-   D0 18     BNE $FA7F    Zut ! On voulait un 2eme zéro
FA67-   C8        INY
FA68-   B1 00     LDA ($00),Y
FA6A-   D0 13     BNE $FA7F    Zut ! On voulait un 3eme zéro
FA6C-   20 52 FA  JSR $FA52    Chic on a trouvé 3 zéros successifs
FA6F-   20 52 FA  JSR $FA52    On met donc à jour le pointeur
FA72-   20 52 FA  JSR $FA52    sur l’octet qui suit ces 3 zéros
FA75-   A5 00     LDA $00      Copie l’adresse de « fin de BASIC »
FA77-   85 9C     STA $9C      dans le pointeur #9C-#9D
FA79-   A5 01     LDA $01      Rappel, le début des variables
FA7B-   85 9D     STA $9D      est situé juste après le BASIC
FA7D-   58        CLI          C’est fini, rétablit les
FA7E-   60        RTS          interruptions et quitte
FA7F-   20 52 FA  JSR $FA52    Pas trouvé : incrémente le pointeur
FA82-   4C 5C FA  JMP $FA5C    et recommence à chercher
FA85-   60        RTS

Après un NEW, les deux premiers octets du début du programme BASIC sont écrasés par une paire de

#00. Pour mieux comprendre la commande OLD, voici un exemple de début de programme BASIC :

10 PRINT « LIGNE 1 »
20 PRINT « LIGNE 2 »

En RAM on a :

Adresse Octet(s) Signification
#0500 00 Drapeau présence d’un programme BASIC
#0501 10 05 Lien, c’est à dire adresse de la ligne suivante, ici #0510
#0503 0A 00 n° de ligne, ici #000A soit 10 en décimal
#0505 BA Code de la commande PRINT
#0506 22 4C 49 47 4E 45 20 31 22 Texte « LIGNE 1 »
#050F 00 Drapeau de fin de la 1ere ligne
#0510 1F 05 Lien, c’est à dire adresse de la ligne suivante, ici #0515
#0512 14 00 n° de ligne, ici #0014 soit 20 en décimal
#0514 BA Code de la commande PRINT
#0515 22 4C 49 47 4E 45 20 32 22 Texte « LIGNE 2 »
#051E 00 Drapeau de fin de la 2eme ligne
#051F 00 00 Lien, ce devrait être le début de la 3eme ligne, mais comme ce lien est
nul, cette paire de zéro indique la fin du programme BASIC
#0521 1E Trace d’un ancien programme ou d’une variable. La valeur de cet octet
est sans intérêt, Mais c’est sur lui qui pointe #9C - #9D qui indique le début des variables. De manière
illogique, cet octet est sauvegardé avec le programme BASIC !

En pratique la fin du programme BASIC est donc marquée par la présence de trois #00 consécutifs.
Après un New, c’est également le cas, puis qu’on trouve trois #00 de #500 à #502, le reste du pro-
gramme étant intact tant qu’on ne coupe pas le courant ou effectue un reset à froid ou met en place un
nouveau programme, d’où l’intérêt de la commande OLD.

TEST DE LA ROM DEPLOMBEUSE

Sauvez un petit programme BASIC en AUTO, par exemple celui en deux lignes indiqué ci-dessus,
avec un CSAVE « ABC »,AUTO

Lorsqu’on charge ce programme avec un Atmos normal, on voit s’afficher deux lignes de texte
« LIGNE 1 » et « LIGNE 2 ». LIST affiche le programme. NEW le fait disparaître et OLD (si vous êtes
sous Sedoric) le fait réapparaître. Si l’on effectue un reset à froid (bouton poussoir du Microdisc) ou
CALL#F88F, le programme disparaît définitivement. Il ne peut plus être récupéré avec un OLD.

Lorsqu’on charge ce même programme avec un Atmos équipé de la ROM déplombeuse, rien ne
s’affiche automatiquement, le programme n’est pas exécuté. LIST indique que le programme est bien
là. Il faut faire un RUN pour l’exécuter. Après un NEW, le programme a disparu, ce qu’on peut tester
avec LIST ou RUN. Mais OLD le fait réapparaître et il peut à nouveau être lancé avec RUN ou sauvé
avec un CSAVE ou un SAVE.
Si l’on effectue un reset à froid (bouton poussoir du Microdisc), le programme disparaît comme le
montre LIST ou RUN, mais il est immédiatement récupéré avec un OLD.

Notez que lors d’un reset à chaud (bouton sous l’Atmos), le programme n’est pas affecté du tout,
tant avec la ROM normale, qu’avec la déplombeuse.

Notez aussi que si des erreurs se produisent lors d’un CLOAD avec la ROM déplombeuse, le mes-
sage fatidique s’affiche pour vous en avertir, mais c’est tout, cela se réduit à un affichage, c’est à
vous de voir s’il y a des corrections à faire.

A suivre…




